


# エクセルマクロ研修 H27版 解答編

問 2-1 次の文字列のうち、変数名として使えるものに○、使えないものに×をつけなさい。

- ・ abcdefg ○アルファベットのみ
- ・ 3.141592 ×先頭が数字
- ・ クラスの平均 ○漢字は使える
- ・ database\_読み込み ○漢字、アルファベット、\_は使える。混在できる。
- ・ 3nen ×先頭が数字
- ・ 三年 ○漢字は使える
- ・ sannen? ×?記号は使えない
- ・ さんねん? ×全角数字・記号は半角に変換される
- ・ jugemujugemugokounosurikirekaijarisuigyonosuigyomatuunraimatufuuraimatu ○72文字

問 2-2 長方形の面積を求める手順を、簡条書きで書きなさい。また、フローチャートを使って書きなさい。変数は、面積  $s$ 、高さ  $h$ 、幅  $w$  とする。


- ①高さ  $h$  を決める
- ②幅  $w$  を決める
- ③縦と横をかける



問 2-3 円の面積を求める手順を、以下の公式を使って簡条書きで書きなさい。また、フローチャートを書きなさい。ただし、円周率は 3.14 とし、変数名は、面積  $s$ 、半径  $r$  とする。

$$\text{円の面積} = \text{円周率} \times \text{半径}^2$$


- ①半径  $r$  を決める
- ②3.14 と  $r$  と  $r$  をかけ、答えを  $s$  に入れる



問 2-4 台形の面積を求める手順を、以下の公式を使って簡条書き及びフローチャートで書きなさい。変数名は、面積  $m$ 、上底  $j$ 、下底  $k$ 、高さ  $t$  とする。

$$\text{面積} = \frac{1}{2} \times (\text{上底} + \text{下底}) \times \text{高さ}$$

- ①上底を決める
- ②下底を決める
- ③高さを決める
- ④上底と下底を足す
- ⑤高さをかける
- ⑥2で割る



問 2-5 クラスの生徒が5名の場合、どのように並べ替えるか考えなさい。

例：1枚ごとに、それまでの束の中に差し込んでいく（挿入ソート）

問 2-6 クラスの生徒が40名の場合、どのように並べ替えるか考えなさい。

例：10の位で分類し、それぞれの束を1の位で並べ替える（ラディックスソート）

問 2-7 並べ替えが終わった後、欠席していた1名が提出した場合、どのように並べ替えるか考えなさい。

例：上から順番に、入るところを探していく（バブルソート）

例：束の中央の番号で2分し、そのどちらの束に入るかを定める。決めた束をさらに2分し、そのどちらに入るかを定める（クイックソート）

問 2-8 問 2-5 の方法で40名のプリントを並べ替えた場合と、問 2-6 の場合とで比べて、どのくらい時間が変わるか考えなさい。

例：枚数による。2-5 は数が多くなると、差し込む場所を探すのが大変。

2-6 は数が少ないときは、最初の段階で分ける束が少ないため、効率的でない。

問 29 次の文章を読み、下の問い（問 a, b）に答えよ。

問 a) 正の整数  $n$  を入力したとき、1 から  $n$  までの数それぞれの 2 乗の和を出力するプログラムを、上の解答群の行を必要なだけ並べて作成せよ。解答群にある行は何回使ってもよい。たとえば 4 を入力すると、 $1 + 4 + 9 + 16 = 30$  なので、「30」と出力する。

問 a) (15 点) 次のいずれか。

アイクオキコソ    イアクオキコソ

採点基準

・イを抜かした:-2 点

・コソをソコとした:-2 点

・ク... コ    の中にケが追加:-3 点

ク... コ    の中にイが追加:-3 点

ク... コ    の中にキが追加:-3 点

など、1つ余分に入っている場合はケアレスミスと考え:-3 点

問 b) 正の整数  $n$  を入力したとき、1 から  $n$  までの数を順番に、奇数については 1 回、偶数についてはその数の回数だけくり返して、出力するプログラムを、上の解答群の行を必要なだけ並べて、次のプログラムを作成せよ。解答群にある行は何回使ってもよい。たとえば 4 を入力すると、「1 2 2 3 4 4 4 4」と出力する。

問 b) (20 点) 次のいずれか。

アクウサエスケセココ    アクエサケセコスシセスコ

アクエシウスケセココ    アクエシセスサケセココ

アクサエケセコスシセスコ    アクサエスシウスケセココ

アクシウスサエスケセココ    アクシセスエサケセココ

アクシセスサエケセココ

採点基準

・「セ」(i を出力) のところを「ソ」(j を出力)にした:-3 点

・「クXコ」とするところを「クX」としている:-3 点

・「ケXコ」とするところを「ケX」としている:-3 点

・「サXス」とするところを「サX」としている:-3 点

・「シXス」とするところを「シX」としている:-3 点

問 3-1 セル C3 に 3 を入力する【プロシージャ名は「Sub ex31」とする。以下同様】。

```
Sub ex31()  
    Sheets("Sheet1").Range("C3").Value = 3  
End Sub
```

問 3-2 セル D3 に 5 を入力する【Sub ex32】。

```
Sub ex32()  
    Sheets("Sheet1").Range("D3").Value = 5  
End Sub
```

問 3-3 セル E3 に、セル C3 の値とセル D3 の値の和を入力する【Sub ex33】。

```
Sub ex33()  
    Range("E3").Value = Range("C3").Value + Range("D3").Value  
End Sub
```

問 3-4 sample5 の実行結果はどのようになるか。

セル D 4 に式「=C1+B2」が設定され、計算結果の 1 1 が表示される。

問 3-5 問 3-3 を Formula メソッドを用いて記述しなさい。

```
Range("E3").Formula = "=C3 + D3"
```

問 3-6 ここまで作成したブックをファイル名「prog1.xlsm」として保存しなさい。

「エクセルマクロ有効ブック」で保存する。

問 3-7 次のプログラムを入力し、実行しなさい。

- a) メッセージボックスで、1, 2, 3, 4, 5 と順に表示される。
- b) メッセージボックスで、1, 3, 5, 7, 9 と順に表示される。
- c) メッセージボックスで、5, 4, 3, 2, 1 と順に表示される。

問 3-8 次の条件を満たす For 文を記述しなさい。また、エラー! 参照元が見つかりません。を参考にしてプログラムを作成・実行し、動作を確認しなさい。

a) 変数 j を 1 から 10 まで繰り返す

```
For j = 1 to 10
```

b) 変数 k を 10 から 100 まで 10 ずつ増やしながら繰り返す

```
For k = 10 to 100 step 10
```

c) 変数 m を 10 から 1 まで繰り返す

```
For m = 10 to 1 step -1
```

d) 変数 n を 100 から 10 まで 7 ずつ減らしながら繰り返す

```
For n = 100 to 10 step -7
```

e) 変数 p を a から b まで繰り返す

```
For p = a to b
```

f) 変数 q をセル B1 の値からセル B2 の値まで繰り返す。ただし、セル B1、B2 の値はともに整数とし、セル B1 の値 < セル B2 の値とする。

```
For q = Range("B1").Value to Range("B2").Value
```


g) 変数 r をセル C1 の値からセル C2 の値まで繰り返す。ただし、セル C1、C2 の値はともに整数とし、セル C1 の値 > セル C2 の値とする。

```
For r = Range("C1").Value to Range("C2").Value step -1
```


問 3-9 次のマクロをそれぞれ入力し、実行結果を比較しなさい。この結果より、値をセルに保存した場合と、変数に保存した場合、どのくらい速度が異なるか評価しなさい。

<pre>Sub speed1()   Dim t As Single   Dim i As Long    t = Timer    For i = 1 To 10000          ' 1万回     Cells(1, 1) = 2     Cells(1, 2) = Cells(1, 1) + i   Next i    MsgBox (Timer - t) &amp; " 秒"  End Sub</pre>	<pre>Sub speed2()   Dim t As Single   Dim i As Long   Dim a As Integer, b As Long    t = Timer    For i = 1 To 100000000      ' 1億回     a = 2     b = a + i   Next i    MsgBox (Timer - t) &amp; " 秒"  End Sub</pre>
--	--

実行例：



speed1



speed2

値をセルに保存した場合：3.59秒÷10000=359<sup>まいくろ</sup>μ秒/回（1回あたり359μ秒）  
 値を変数に保存した場合：2.18秒÷100000000=0.0218μ秒/回  
 よって、変数に保存したほうが、約 16,467 倍速い

問 3-10 次の繰り返しを実現するための Do 文を記述しなさい。

- セルA1の値がゼロでない間は繰り返す  
 Do While Range("A1").Value <> 0 または  
 Do Until Range("A1").Value = 0
- セルA1の値がプラスの間は繰り返す  
 Do While Range("A1").Value > 0 または  
 Do Until Range("A1").Value <= 0
- セルB1の値がゼロになるまで繰り返す  
 Do Until Range("B1").Value = 0 または  
 Do While Range("B1").Value <> 0
- セルB1の値が100を超えるまで繰り返す  
 Do While Range("B1").Value <= 100 または  
 Do Until Range("B1").Value > 100
- セルC1の値が100以下かつセルD1の値が100以下の間は繰り返す  
 Do While Range("C1").Value <= 100 and Range("D1").Value <= 100  
 Do Until Range("C1").Value > 100 or Range("D1").Value > 100

問 3-11 次の条件分岐を実現するための If 文を記述しなさい。

a) 変数bの値がゼロならば、変数cに3を代入する

`If b = 0 Then c = 3`

b) 変数dの値がゼロでないならば、変数eにd×10を代入する

`If d <> 0 Then e = d * 10`

c) セルB2の値がゼロならば、変数fにセルB2の値を代入する

`If Range("B2").Value = 0 Then f = Range("B2").Value`

d) セルB3の値がゼロより大きいならば、変数gにセルB2の値を代入する

`If Range("B3").Value > 0 Then g = Range("B2").Value`

e) セルB4の値が30未満ならば、変数hに変数h+1の値を代入する

`If Range("B4").Value < 30 Then h = h + 1`

f) セルB2から下にp、右に0移動したセルの値がゼロでないならば、変数sにその値を加える

`If Range("B2").Offset(p, 0).Value <> 0 Then s = s + Range("B2").Offset(p, 0).Value`

問 3-12 sample8 を変更して、鹿沼市以外の出身者の数を数えるマクロを 10 秒で作成しなさい。

`If Range("J5").Offset(i, 0).Value = "鹿沼市" Then` …②

を、

`If Range("J5").Offset(i, 0).Value <> "鹿沼市" Then` …②

に変更する。

問 4-1 B 列の番号順に並べ替えるマクロ sample12 を作成しなさい。

```
Sub sample12()  
  
    Sheets("基礎データ").Range("b6:x10").Sort _  
        Key1:=Range("b6"), Order1:=xlAscending, Orientation:=xlSortColumns  
  
End Sub
```

問 5-1 11 ページから 15 ページまで、40 部、ページ単位(11 ページを 40 部、12 ページを 40 部 …の順番)で印刷したい場合、どのような記述となるか考えなさい。

`ActiveWindow.SelectedSheets.PrintOut From:=11, To:= 15, Copies:=40, Collate:=False`

問 5-2 上記のプロシージャを実行するためのボタンを設定しなさい。

4.2 を参考に、ボタンを配置し、実行したいマクロとして「印刷」を指定。

問 6-1 何枚のシートまで結合できるか試しなさい。

適当なシートのコピーを繰り返し、何枚まで作成可能かを実行して調べる。

シート結合の途中経過

10	エクセルのマクロは経験ありますか?	ない
11	32×2の答えはどれ? 34, 48, 64, 98	34
12	地球ができたのはおよそ何年前?	46年前

少なくとも、2,620 枚までは可能。仕様上は、メモリの容量による。

問 6-2 集計表（基礎データ）の行列を反転せず、そのまま貼り付けて利用するマクロを作成しなさい（すなわち、全員分のデータが、縦に並ぶ）。

```

Sub 表の作成 ()
    Dim i As Integer          ' 繰り返し変数、何番目を処理しているか
    Dim syozoku As String     ' 所属名
    Dim namae As String       ' 氏名

    Sheets("基礎データ").Select
    Range("c5:z100").Value = "" ' 集計先シートの選択
                                   ' 集計先セルのクリア

    For i = 1 To Range("人数").Value ' 人数分繰り返し返す
        syozoku = Sheets("t" & i).Range("c6").Value ' 所属を取得
        namae = Sheets("t" & i).Range("e6").Value ' 氏名を取得

        Sheets("基礎データ").Range("b5").Offset(0, i).Value = i ' 番号を集計先に転記
        Sheets("基礎データ").Range("b5").Offset(1, i).Value = syozoku ' 所属を集計先に転記
        Sheets("基礎データ").Range("b5").Offset(2, i).Value = namae ' 氏名を集計先に転記

        Sheets("t" & i).Select ' 転送元シートの選択
        Range("e9:e28").Copy ' データの範囲をコピー

        Sheets("基礎データ").Select ' 集計先シートの選択
        Sheets("基礎データ").Range("b5").Offset(3, i).Select ' 集計先セルの選択
        Selection.PasteSpecial Paste:=xlPasteFormulas, _ ' 集計先シートへ貼り付け
            Operation:=xlNone, SkipBlanks:=False, Transpose:=False

    Next i ' 繰り返しここまで

End Sub

```

	A	B	C	D	E	F	G	H
1								
2		回答人数	5					
3								
4								
5		番号	1	2	3	4	5	
6		所属	01小学校	02中学校	03高等学校	04特別支援学	05小学校	
7		氏名	01 一郎	02 花子	03 三郎	04 四朗子	05 吾郎	
8		問1	1	2	1		1	
9		問2	小学校	中学校	高校	特別支援	その他	
10		問3	国語	数学	工業	社会	情報	
11		問4	がんばります。	楽しみです。	ついていければ	気合はいつま	気楽にいぎます。	
12		問5	はい	いいえ	いいえ	はい	はい	
13		問6	鹿沼市	宇都宮市	那須塩原市	鹿沼	栃木市	
14		問7	100000	20000	百万円	25300	70000	
15		問8	りんご	りんご	みかん	りんご	みかん	
16		問9	3	0	10	8	4	
17		問10	ない	ない	ある	ある	ない	
18		問11	34	48	64	98	64	
19		問12	46年前	46万年前	46億年前	46兆年前	46万年前	
20		問13	100	110	500	141	1000	
21		問14	N	E	W	S	W	
22		問15	1	2	8	16	2	
23		問16	○	×	○	×	○	
24		問17	<	=	>	<	>	
25		問18	2	4	0	1	3	
26		問19	○	×	○	×	○	
27		問20	seven	eight	nine	eleven	twelve	
28								
29								

問 8-1 Rnd 関数は、0 以上 1 未満の乱数値を返す。0 ～ 9 の整数の乱数値を得るためには、どのような式を用いればよいか考えなさい。

$$\text{Int}(\text{Rnd}() * 10)$$

問 8-2 1 ～ 6 の整数の乱数値を得るためには、どのような式を用いればよいか考えなさい。

$$\text{Int}(\text{Rnd}() * 6) + 1$$

問 8-3 101~999 の整数の乱数値を得るためには、どのような式を用いればよいか考えなさい。

$\text{Int}(\text{Rnd}() * (999 - 101 + 1)) + 101$

問 8-4 1~10 の乱数を 10 個生成し、その度数分布 (出現回数) を求めよ。乱数を 100 個、1,000 個と増やした場合、度数分布はどのようにになるか。ヒント: 数を数えるために、配列 a(10) を使うとよい。

度数分布の例:

1	0
2	0
3	2
4	0
5	0
6	2
7	1
8	0
9	2
10	3

100回

1	12
2	13
3	10
4	10
5	12
6	12
7	11
8	7
9	3
10	10

1000回

1	98
2	89
3	116
4	105
5	103
6	102
7	96
8	96
9	104
10	91

10000回

```

Sub test()
  Dim r As Integer
  Dim i As Integer
  Dim a(10) As Integer

  For i = 1 To 10                                '回数を数える配列をゼロにする
    a(i) = 0
  Next i

  For i = 1 To 1000                               '1000回繰り返す
    r = Int(Rnd() * 10) + 1                       '乱数を作る
    a(r) = a(r) + 1                               '出現した数を1回増やす
  Next i

  For i = 1 To 10                                '結果をエクセルの表に埋める
    Range("a1").Offset(i, 0).Value = i
    Range("a1").Offset(i, 1).Value = a(i)
  Next i

End Sub

```

問 8-5 1～10の乱数2つ生成し、その和を求めよ。さらに、この乱数を1,000回発生させたときの度数分布を求めよ。どのような分布になっているか。

中央が多く、両端が少ない。正規分布。

1	0
2	5
3	20
4	25
5	46
6	44
7	76
8	65
9	83
10	87
11	95
12	78
13	80
14	74
15	54
16	54
17	43
18	42
19	18
20	11

```
Sub test2()  
  Dim r1 As Integer, r2 As Integer  
  Dim i As Integer  
  Dim a(20) As Integer  
  
  For i = 1 To 10  
    a(i) = 0  
  Next i  
  
  For i = 1 To 1000  
  
    r1 = Int(Rnd() * 10) + 1  
    r2 = Int(Rnd() * 10) + 1  
  
    a(r1 + r2) = a(r1 + r2) + 1  
  
  Next i  
  
  For i = 1 To 20  
    Range("a1").Offset(i, 0).Value = i  
    Range("a1").Offset(i, 1).Value = a(i)  
  Next i  
  
End Sub
```

問 8-6 User32.dllには、ここで使用した GetAsyncKeyState 以外にどのような関数が含まれているか調べなさい。

参照：<http://www.rinku.zaq.ne.jp/ultimate/lecture/api/user32.htm>

EnableWindow, ExitWindowsEx, GetClipboardData, GetWindowRect...

問 8-7 User32.dll 以外にどのような dll が存在するか調べなさい。

参照：上記

Gdi32, Shell32, Wsock32...

問 8-8 次のプログラムを入力し、実行しなさい（実行時には、エクセルのシート内容の変化に注意すること。また、キーボードの左矢印、右矢印を押してみること）。

洞窟ゲーム（別名スキーゲーム）ができる。

問 8-9 座標、大きさが異なる四角形を 100 個描くマクロを作成しなさい。

実行例



```

Sub graphics_rect()
    Dim ch As Chart          ' 図オブジェクト
    Dim a As Shape          ' 図形オブジェクト
    Dim i As Integer       ' 繰り返し用変数
    Dim x1 As Integer, y1 As Integer ' 画面上の座標
    Dim x2 As Integer, y2 As Integer ' 画面上の座標


    Set ch = Charts.Add    ' シートに図オブジェクトを追加

    For i = 1 To 100      ' 100回繰り返す
        x1 = Int(Rnd() * 600) ' x座標を乱数(0~599)で求める
        y1 = Int(Rnd() * 400) ' y座標を乱数(0~399)で求める
        x2 = Int(Rnd() * 600) ' x座標を乱数(0~599)で求める
        y2 = Int(Rnd() * 400) ' y座標を乱数(0~399)で求める
        Set a = ch.Shapes.AddShape(msoShapeRectangle, x1, y1, x2, y2) ' 四角を図形として追加する
        DoEvents          ' OSに動作を委譲する(描画する)
    Next i
End Sub

```

問 8-10 適当な座標を持つ直線を色を変えて 100 個描くマクロを作成しなさい。

実行例



```

Sub graphics_line()
    Dim ch As Chart          ' 図オブジェクト
    Dim a As Shape          ' 図形オブジェクト
    Dim i As Integer       ' 繰り返し用変数
    Dim x1 As Integer, y1 As Integer ' 画面上の座標
    Dim x2 As Integer, y2 As Integer ' 画面上の座標
    Dim r As Byte, g As Byte, b As Byte ' 色

    Set ch = Charts.Add    ' シートに図オブジェクトを追加


    For i = 1 To 100      ' 100回繰り返す
        x1 = Int(Rnd() * 600) ' x座標を乱数(0~599)で求める
        y1 = Int(Rnd() * 400) ' y座標を乱数(0~399)で求める
        x2 = Int(Rnd() * 600) ' x座標を乱数(0~599)で求める
        y2 = Int(Rnd() * 400) ' y座標を乱数(0~399)で求める
        r = Int(Rnd() * 256) ' 色の赤要素を乱数(0~255)で求める
        g = Int(Rnd() * 256) ' 色の緑要素を乱数(0~255)で求める
        b = Int(Rnd() * 256) ' 色の青要素を乱数(0~255)で求める
        Set a = ch.Shapes.AddLine(x1, y1, x2, y2) ' 直線を図形として追加する
        a.Line.ForeColor.RGB = RGB(r, g, b) ' 輪郭線の色を設定する
        DoEvents          ' OSに動作を委譲する(描画する)
    Next i
End Sub

```


問 8-11 次のマクロを入力し、実行しなさい。

```
Sub g2()  
  Dim ch As Chart          ' 図オブジェクト  
  Dim a As Shape          ' 図形オブジェクト  
  Dim i As Integer        ' 繰り返し用変数  
  
  Set ch = Charts.Add     ' シートに図オブジェクトを追加  
  
  For i = 0 To 300 Step 5  
    Set a = ch.Shapes.AddLine(i, 0, 0, 300 - i)  
    Set a = ch.Shapes.AddLine(300, i, 300 - i, 300)  
    DoEvents  
  Next i  
  
End Sub
```

実行結果



問 8-12 分数の計算：次のような流れのパワーポイントを作成しなさい。



省略

問 9-1 文字列型変数 b を 11 個確保するための命令を記述しなさい。

**Dim b(11) As String**

問 9-2 倍精度型浮動小数点数型変数 c を 10,000 個確保するための命令を記述しなさい。

**Dim c(10000) As Double**